

سخت افزار کنترل فاصله برای ربات شیلد به کمک ماژول SRF04

سجاد باغستانی [لینک پروژه](#)

مقدمه

در این پروژه قصد داریم سخت افزاری برای اندازه گیری اجسام نزدیک به ربات شیلد به کمک ماژول التراسونیک SRF04 طراحی کنیم. این پروژه شامل دو بخش طراحی سخت افزار شامل انتخاب قطعات، نحوه ی اتصال، جانمایی و طراحی نرم افزار شامل برنامه نویسی های مربوطه، تست و اجرا می باشد. به دلیل تغییرات ساده تر در بخش نرم افزار مقل جابجایی پین ها به صورت نرم افزار و حساسیت بیشتر سخت افزار در ادامه ابتدا سخت افزار پروژه طراحی شده و سپس به سراغ نرم افزار می رویم.

طراحی سخت افزار


برای بخش سخت افزاری پروژه شامل مدارات و جانمایی از نرم افزار Altium Designer و همچنین برای افزایش سرعت در افزودن قطعات به پروژه از افزونه Altium Library Loader استفاده شده است.

انتخاب میکروکنترلر

به جهت هزینه ی کمتر، امکان مونتاژ به کمک ربات از نسخه ی SMD میکروکنترلر ATMEGA328P-AU اسفاده شده است.

مدار ریست


از آنجایی که ریست میکروکنترلر به صورت Active-Low صورت می گیرد. مدار زیر شامل یک push-button و یک مقاومت pull-up می باشد، در نظر گرفته شده است.



شکل ۱- مدار reset

مدار پروگرامر


برای پروگرام کردن میکروکنترلر از یک مبدل USB به Serial(UART) آی سی CH340 و یک متصل کننده ی USB مشابه شکل زیر در نظر گرفته شده است که امکان پروگرام کردن میکروکنترلر را به کمک USB و Arduino IDE فراهم می کند.



شکل ۲- مدار پروگرامر

مدار تغذیه


برای تغذیه ی سخت افزار یک باتری ۱۲ ولتی در نظر گرفته شده است و به دلیل ۵ ولت بودن تغذیه ی بقیه ی قطعات اکتیو شامل میکروکنترلر و ... نیاز به یک رگولاتور ۱۲ به ۵ ولت داریم که از AMS1117 به همراه خازن های جانبی که برای کاهش نویز منبع و جلوگیری از اثرات مخرب افت ناگهانی ولتاژ در ورودی و حفظ پایداری و کاهش ریبیل ولتاژ در خروجی مطابق شکل ۳ در نظر گرفته شده است.



شکل ۳- مدار تغذیه

مدار سطح ولتاژ باتری


به جهت خوانش سطح ولتاژ باتری در نظر داریم که از مبدل آنالوگ به دیجیتال میکروکنترلر استفاده کنیم اما به دلیل وجود اختلاف ۱۲ ولت باتری و ۵ ولت بودن سطح ولتاژ مبدل ADC ابتدا به کمک یک تقسیم مقاومت مطابق شکل ۴ سطح ولتاژ را تغییر داده و سپس به کمک نرم افزار آنرا جهت نمایش به کاربر اصلاح می کنیم.




شکل ۴- خوانش ولتاژ باتری

سایر سخت افزار


در ادامه مطابق خواسته ی مساله از ۵ عدد push-button با pull-up داخلی جهت تغییر مد سخت افزار، یک عدد Buzzer برای هشدار که از مدل Active با جریان کم (بدون نیاز به مدار جانبی)، یک عدد OLED LCD SSD1306 که با رابط I2C به میکروکنترلر ارتباط می گیرد و سنسور SRF04 التراسونیک جهت اندازه گیری فاصله استفاده شده است. شماتیک سخت افزار، جانمایی و شکل سه بعدی مدار به ترتیب در شکل های ۵-۷ قابل مشاهده است.



شکل ۵- شماتیک سخت افزار



شکل ۶- جانمایی ۲ بعدی



شکل ۷- جانمایی ۳ بعدی

طراحی نرم افزار

برای بخش نرم افزار از Arduino IDE 1.8.19 استفاده شده است. مطابق خواسته ی مساله ابتدا یک کتابخانه به منظور خواندن دیتای سنسور و متد های مختلف آماده شده و سپس به کمک توابع مختلف سخت افزار کنترل می شود.

کتابخانه ی سنسور SRF04

برای ساختن کتابخانه ابتدا دو فایل با پسوند های `.h` برای تعریف کلاس مربوطه و متغیر ها و `.cpp` برای عملکرد کلاس شامل نحوه ی ساختن شیء و عمل کردن توابع مربوطه ساخته می شود.

فایل `.h` مطابق زیر تعریف شده است که شامل `include guards` و افزودن آردوئو و تعریف کلاس می باشد.

```
1. #ifndef ULTRASONICSRF04_H
2. #define ULTRASONICSRF04_H
3.
4. #include <Arduino.h>
5.
6. class UltrasonicSRF04 {
7.     private:
8.         uint8_t trigPin;
9.         uint8_t echoPin;
10.
11.     public:
12.         UltrasonicSRF04(uint8_t trig, uint8_t echo);
13.         float getDistance(String unit = "cm");
14.         bool inBound(float Bound);
15.         float getFilteredDistance();
16. };
17.
18. #endif
19.
```

سپس به پیاده سازی نحوه ی عملکرد سنسور در فایل `.cpp` می پردازیم. در ابتدا نیاز به `constructor` داریم که هنگام صدا زدن شیء سنسور را مطابق با تعریف انجام شده که تنظیم کردن پین های ارسال و دریافت است داریم. سپس، مطابق دیتاشیت سنسور التراسونیک برای اندازه گیری فاصله نیاز به ارسال یک پالس و اندازه گیری زمان پالس دریافتی می باشد که باتوجه به زمان پالس دریافتی و سرعت صوت فاصله به صورت زمان دریافتی تقسیم بر ۵۸ و یا زمان دریافتی تقسیم بر ۱۴۶ به ترتیب فاصله بر حسب سانتی متر یا اینچ مشخص می گردد که در تابع `getDistance` پیاده شده است. در متد دیگر خواسته شده است که در صورت وجود جسم در بازه ی مشخص شده یک متغیر `true` و `false` برگردانده شده که در تابع `inbound` پیاده شده است و در نهایت برای افزایش دقت سنسور در مواردی که سنسور روی جسم متحرک می باشد از متد میانگین گیری در تابع `getFilteredDistance` پیاده شده است. در این متد به دلیل ثابت بودن فاصله و متغیر بودن نویز هنگام میانگین گیری مقدار نویز به طور متوسط بر تعداد دفعات اندازه گیری تقسیم شده اما سیگنال اصلی ثابت خواهد ماند.

```

1. #include "UltrasonicSRF04.h"
2.
3. UltrasonicSRF04::UltrasonicSRF04(uint8_t trig, uint8_t echo) {
4.   trigPin = trig;
5.   echoPin = echo;
6.
7.   pinMode(trigPin, OUTPUT);
8.   pinMode(echoPin, INPUT);
9. }
10.
11. float UltrasonicSRF04::getDistance(String unit) {
12.   digitalWrite(trigPin, LOW);
13.   delayMicroseconds(2);
14.   digitalWrite(trigPin, HIGH);
15.   delayMicroseconds(10);
16.   digitalWrite(trigPin, LOW);
17.
18.   long duration = pulseIn(echoPin, HIGH, 30000);
19.
20.   if (duration == 0) return -1;
21.
22.   if (unit == "inch") {
23.     return duration / 148.0;
24.   } else {
25.     return duration / 58.0;
26.   }
27. }
28.
29. bool UltrasonicSRF04::inBound(float bound) {
30.   return getDistance() <= bound;
31. }
32.
33. float UltrasonicSRF04::getFilteredDistance(){
34.   float distance = 0;
35.   int trueSample = 0;
36.   for(int i=0; i < 5; i++){
37.     float newDistance = getDistance();
38.     if(newDistance >= 0){
39.       distance = distance + newDistance;
40.       trueSample++;
41.     }
42.     delay(100);
43.   }
44.   if(trueSample == 0){
45.     return -1;
46.   } else{
47.     return distance / trueSample;
48.   }
49.
50.
51. }
52.

```

پیکربندی فایل اصلی

پس از نهایی کردن کتابخانه ی مربوط به راه اندازی و استفاده ی سنسور التراسونیک نیاز به توسعه ی پیکره ی اصلی نرم افزار به کمک Arduino IDE در فایل .ino می باشد که در ادامه مرحله به مرحله به توضیح آن پرداخته خواهد شد.

افزودن کتابخانه ها

به منظور استفاده از نمایشگر OLED SSD1306، کنترل سنسور التراسونیک و ارتباط lcd و میکروکنترلر نیاز به کتابخانه ها مربوطه می باشد که مطابق با زیر تعریف شده اند. (از آنجایی کتابخانه ی سنسور التراسونیک به منظور تست در وبسایت wokwi به صورت محلی تعریف می شود کتابخانه ی مربوطه بین "" نیز تعریف شده است).

```
1. #include <Wire.h>
2. #include <Adafruit_GFX.h>
3. #include <Adafruit_SSD1306.h>
4. #include <UltrasonicSRF04.h>
5. //include "UltrasonicSRF04.h"
6.
```

تعریف ها و متغیرها

در این بخش به منظور حفظ پیوستگی کد نویسی و تغییرات ثانویه ی ساده تر متغیر ها پیش پردازش ها و متغیر های ثابت مطابق با زیر انجام می شود. که شامل طول و عرض lcd، پین ورودی آنالوگ جهت خوانش سطح ولتاژ باتری، مقادیر ثابت مقاومت برای تغییر سطح ولتاژ از ۱۲ به ۵ ولت و مد های مختلف حالت می باشد.

```
1. #define SCREEN_WIDTH 128
2. #define SCREEN_HEIGHT 64
3. #define BUZZER_PIN 24
4.
5. const int BATTERY_PIN = A0;
6. const float R1 = 4700.0;
7. const float R2 = 10000.0;
8. const int MODE_BUTTONS[] = {12, 13, 14, 15, 16};
9. const int MODE_NUMBERS = 5
10.
;
```

ساخت شیء

نیاز به ساخت شیء از کلاس مربوطه و سپس کنترل آنها می باشد که مطابق با OLED LCD برای استفاده از سنسور التراسونیک و زیر انجام شده است.

```
1. Adafruit_SSD1306 display(SCREEN_WIDTH, SCREEN_HEIGHT, &Wire, -1);
2. UltrasonicSRF04 sensor(10, 9);
3.
```

تنظیمات و راه اندازی OLED LCD

به منظور راه اندازی اولیه lcd ابتدا نیاز به تعریف lcd می باشد که مطابق زیر انجام شده است و سپس در مراحل جداگانه جهت نمایش مقادیر مختلف کنترل خواهد شد. ابتدا ارتباط میکروکنترلر با lcd چک شده و در صورت عدم ارتباط یه پیغام به ترمینال مبنی بر عدم شناسایی lcd و در صورت شناسایی پیغامی را مبنی بر شناسایی lcd روی خود lcd نمایش می دهد.

```
1. void setOled(){
2.   if (!display.begin(SSD1306_SWITCHCAPVCC, 0x3C)){
3.     Serial.println("OLED LCD Not Found");
4.     while(true);}
5.
6.   display.clearDisplay();
7.   display.setTextSize(1);
8.   display.setTextColor(SSD1306_WHITE);
9.   display.setCursor(0,0);
10.  display.println("HELLO, OLED LCD Found");
11.  display.display();
12. }
13.
```

تنظیمات خواندن باتری و نمایش سطح ولتاژ

مطابق با خواسته ی مساله مبنی بر نمایش سطح ولتاژ باتری ۱۲ ولت که پس از تقسیم مقاومت، به سطح ۵ ولت رسیده است، از طریق پین آنالوگ A0 خوانده شده و مقدار آن از سطح ۵ ولت مشخص شده و مجدداً به سطح ۱۲ ولت در تابع batteryStatus تبدیل می شود. این مقدار برگردانده شده به کمک تابع showBatteryStatus در صورت نیاز (مد پیشفرض صفر) روی نمایشگر نمایش داده می شود.

```
1. float batteryStatus(){
2.   int adc_val = analogRead(BATTERY_PIN);
3.   float vout = adc_val * 5.0 / 1023;
4.   float vout_bat = vout * (R1 + R2)/R2;
5.   return vout_bat;
6. }
7.
8. void showBatteryStatus(){
9.   float battery = batteryStatus();
10.  display.clearDisplay();
11.  display.setTextSize(1);
12.  display.setTextColor(SSD1306_WHITE);
13.  display.setCursor(0,0);
14.  display.print("Battery: ");
15.  display.print(battery);
16.  display.println(" V");
17.  display.display();
18.
19. }
20.
```


نمایش فاصله

جهت نمایش فاصله از تابع `showDistance` استفاده شده است که با دریافت مقدار فاصله از جسم و واحد اندازه گیری، آنرا روی نمایشگر، نمایش می دهد.

```
1. void showDistance(float distance, String unit) {
2.   display.clearDisplay();
3.   display.setTextSize(1);
4.   display.setTextColor(SSD1306_WHITE);
5.   display.setCursor(0, 0);
6.   display.print("Distance: ");
7.   display.print(distance);
8.   display.print(" ");
9.   display.println(unit);
10.  display.display();
11. }
12.
```

تنظیمات بازر

تنظیمات بازر درون تابع `setBuzzer` و نحوه ی به صدا درآمدن آن در تابع `alarm` مطابق زیر توسعه داده شده است.

```
1. void setBuzzer(){
2.   pinMode(BUZZER_PIN, OUTPUT);
3.   digitalWrite(BUZZER_PIN, LOW);
4. }
5.
6. void alarm(bool cond){
7.   if (cond){
8.     digitalWrite(BUZZER_PIN, HIGH);}else{
9.     digitalWrite(BUZZER_PIN, LOW);}
10. }
11.
```

تغییر Mode

طبق خواسته ی مساله ۵ کلید `push-button` جهت تغییر مد کاربری ربات تعبیه شده است. تابع `setButtons` تنظیمات مربوط به کلید ها به صورت `push-pull` را انجام داده تابع `buttonStatus` وضعیت فشردن کلید را بررسی کرده و برمیگرداند و در نهایت تابع `selectMode` وضعیت ربات را مشخص می کند. لازم به ذکر است متغیر `mode` برای حفظ مد در نظر گرفته شده است که به صورت پیشفرض صفر می باشد. مد های در نظر گرفته شده به صورت زیر می باشد:

```
1. 0 -> Battery Status
2. 1 -> Show Distance(cm)
3. 2 -> Show Distance(inch)
4. 3 -> Show Filtered Distance
5. 4 -> Show Battery Status and Show Distance(cm)
```

```

1. void setButtons(){
2.   for(int i = 0; i< MODE_NUMBERS; i++){
3.     pinMode(MODE_BUTTONS[i], INPUT_PULLUP);
4.   }
5. }
6.
7. bool buttonStatus(int ind){
8.   return digitalRead(MODE_BUTTONS[ind]);
9. }
10. int mode = 0;
11. void selectMode(){
12.   for (int i=0; i<5; i++){
13.     if (!buttonStatus(i)){
14.       mode = i;
15.       break;
16.     }
17.   }
18. }
19. void showCurrentMode(){
20.   switch(mode){
21.     case 0:
22.       showBatteryStatus();
23.       break;
24.     case 1:
25.       showDistance(sensor.getDistance(), "cm");
26.       break;
27.     case 2:
28.       showDistance(sensor.getDistance("inch"), "inch");
29.       break;
30.     case 3:
31.       showDistance(sensor.getFilteredDistance(), "cm");
32.       break;
33.     case 4:
34.       showBatteryStatus();
35.       delay(1000);
36.       showDistance(sensor.getDistance(), "cm");
37.   }
38. }
39.

```

تنظیمات نهایی

در نهایت توابعی که در بالا به صورت کاملاً مازولار طراحی شد را در دو تابع `setup` و `loop` صدا زده تا ابتدا تنظیمات اولیه ی آنها انجام شده و سپس به ترتیب گفته شده اجرا شوند.



```

1. void setup() {
2.   Serial.begin(9600);
3.   setOled();
4.   setBuzzer();
5.   setButtons();
6. }
7.
8. void loop() {
9.   selectMode();
10.  showCurrentMode();
11.  bool isInBound = sensor.inBound(30.0);
12.  alarm(isInBound);
13.  delay(500);
14. }

```

تست نهایی

به منظور آزمایش نهایی قسمت بازر به کمک یک led در وبسایت wokwi تست شد که در تصویر ۸ نتایج آن قابل مشاهده است. به هنگام فاصله ی بیشتر از ۳۰ سانتی متر led در وضعیت خاموش (۸-الف) و فواصل کمتر از ۳۰ سانتی متر در حالت روشن (۸-ب) قرار دارد.



شکل ۷- تست بازر

تمامی فایل های مربوط به طراحی سخت افزار، نرم افزار و تست نهایی در [لینک گیت هاب](#) پروژه موجود است.